MÉTODOS DE SIMULACIÓN Y MODELADO

Solución al examen de Febrero 2026

Pregunta 1 (2 puntos)

1.a (1 punto) Explique cómo se define el *índice* de un sistema de ecuaciones algebraico diferenciales (DAE).

1.b (1 punto) Ponga un ejemplo de un sistema DAE de índice 2, demostrando que el sistema tiene índice 2.

Solución a la Pregunta 1

Véanse las páginas 318 a 320 del texto base.

Pregunta 2 (2 puntos)

A continuación se muestra un modelo matemático, donde la variable t representa el tiempo.

$$\begin{aligned}
\frac{dx_1}{dt} - a_2 \cdot \frac{dx_2}{dt} &= b_2 \\
b_1 \cdot x_1 + x_2 \cdot x_3 &= b_2 \cdot \frac{dx_2}{dt} \\
x_2 &= b_1 \cdot x_1 - \frac{dx_1}{dt} \\
x_3 &= x_4 - x_1
\end{aligned}$$

$a_1 = 7.65 \cdot 10^{-3}$	$a_2 = \frac{2}{3} \cdot a_1$
$b_1 = 4.07 \cdot 10^{-2}$	$b_2 = b_1 \cdot t$

2.a (1 punto) Clasifique las variables del modelo en parámetros, variables de estado y variables algebraicas. A continuación, realice la asignación de la causalidad computacional. Explique detalladamente el procedimiento seguido para ello.

2.b (1 punto) Escriba el diagrama de flujo del algoritmo para la simulación de este modelo. Emplee el método de integración de Euler explícito. Asigne el valor inicial que desee a las variables de estado. La condición de finalización de la simulación es que el tiempo simulado alcance el valor 10 s.

Solución a la Pregunta 2

Las variables del modelo pueden clasificarse de la forma siguiente:

– Parámetros y constantes: a_1, a_2, b_1
– Variables de estado: x_1, x_2
– Variables algebraicas: b_2, x_3, x_4

Para asignar la causalidad computacional al modelo, se sustituyen las derivadas de las variables de estado por variables mudas:

$$\frac{dx_1}{dt} \to derx1, \qquad \frac{dx_2}{dt} \to derx2$$

Omitiendo las ecuaciones en las que se asigna valor a los parámetros y las constantes, se obtiene el modelo siguiente:

$$\begin{aligned}
derx1 - a_2 \cdot derx2 &= b_2 && (1.1)\\
b_1 \cdot x_1 + x_2 \cdot x_3 &= b_2 \cdot derx2 && (1.2)\\
x_2 &= b_1 \cdot x_1 - derx1 && (1.3)\\
x_3 &= x_4 - x_1 && (1.4)\\
b_2 &= b_1 \cdot t && (1.5)
\end{aligned}$$

Las variables del modelo pueden clasificarse en conocidas (el tiempo, los parámetros, las constantes y las variables de estado) y desconocidas (las variables algebraicas y las derivadas de las variables de estado):

– Conocidas: t
a_1, a_2, b_1
x_1, x_2
– Desconocidas: x_3, x_4, b_2
$derx1$, $derx2$

Con el fin de analizar si el modelo es *estructuralmente singular*, se comprueba que:

1. El número de ecuaciones y de variables desconocidas (incógnitas) es el mismo. Este modelo tiene 5 ecuaciones y 5 incógnitas: x_3, x_4, b_2, $derx1$, $derx2$.

2. Cada incógnita puede emparejarse con una ecuación en que aparezca y con la cual no se haya emparejado ya otra incógnita.

$$
\begin{array}{lll l}
derx2 & \rightarrow & derx1 - a_2 \cdot derx2 = b_2 & \text{Ec. (1.1)} \\
x_3 & \rightarrow & b_1 \cdot x_1 + x_2 \cdot x_3 = b_2 \cdot derx2 & \text{Ec. (1.2)} \\
derx1 & \rightarrow & x_2 = b_1 \cdot x_1 - derx1 & \text{Ec. (1.3)} \\
x_4 & \rightarrow & x_3 = x_4 - x_1 & \text{Ec. (1.4)} \\
b_2 & \rightarrow & b_2 = b_1 \cdot t & \text{Ec. (1.5)}
\end{array}
$$

Finalmente, las ecuaciones del modelo, con la asignación de la causalidad computacional señalada, son las siguientes:

$$
\begin{array}{rcl}
derx1 - a_2 \cdot [derx2] & = & b_2 \\
b_1 \cdot x_1 + x_2 \cdot [x_3] & = & b_2 \cdot derx2 \\
x_2 & = & b_1 \cdot x_1 - [derx1] \\
x_3 & = & [x_4] - x_1 \\
[b_2] & = & b_1 \cdot t
\end{array}
$$

El modelo ordenado y resuelto es el siguiente (existen otras posibles ordenaciones de las ecuaciones que son igualmente válidas):

$$
\begin{array}{rcl}
x_1, x_2 & & \text{Variables de estado} \\
[b_2] & = & b_1 \cdot t \\
[derx1] & = & b_1 \cdot x_1 - x_2 \\
[derx2] & = & \dfrac{derx1 - b_2}{a_2} \\
[x_3] & = & \dfrac{b_2 \cdot derx2 - b_1 \cdot x_1}{x_2} \\
[x_4] & = & x_3 + x_1
\end{array}
$$

En la figura siguiente se muestra el diagrama de flujo para la simulación del modelo, empleando el método de integración de Euler explícito y considerando como condición de finalización que el tiempo simulado alcance el valor 10 s. Se han asignado valores arbitrarios a los valores iniciales de las variables de estado.



Inicio
Asignar valor al incremento en el tiempo (Δt)
Δt = 0.01
Inicializar la variable tiempo
t = 0
Asignar valor a los parámetros
$a_1=7.65\cdot 10^{-3}$, $a_2=2/3\cdot a_1$, $b_1=4.07\cdot 10^{-2}$
Asignar valor inicial a la variable de estado
$x_1(0)=10$, $x_2(0)=1$
Calcular el valor, en el instante t, de las variables algebraicas y de las derivadas
$b_2(t) = b_1\cdot t$
$derx1(t) = b_1\cdot x_1(t) - x_2(t)$
$derx2(t) = (derx1(t) - b_2(t)) / a_2$
$x_3(t) = (b_2(t)\cdot derx2(t) - b_1\cdot x_1(t)) / x_2(t)$
$x_4(t) = x_3(t) + x_1(t)$
t=t+Δt
Incrementar la variable tiempo
t > 10
sí
Fin
no
Calcular el valor, en instante t+Δt, de la variable de estado
$x_1(t+\Delta t) = x_1(t) + derx1(t)\cdot \Delta t$
$x_2(t+\Delta t) = x_2(t) + derx2(t)\cdot \Delta t$

Pregunta 3 (4 puntos)

A continuación se muestra un modelo descrito en lenguaje Modelica y preguntas acerca de él.

```
model Pregunta3
  parameter Real p1(start=0, fixed=false);
  parameter Real p2 = 11.2;
  Real x1 (start=-1,  fixed=true);
  Real x2 (start=3,   fixed=false);
  Real y1 (start=20,  fixed=true);
  Real y2 (start=1.2, fixed=true);
  Integer n1;
  Integer n2(start=-10,  fixed=true);
  Boolean b (start=true, fixed=true);
equation
  y2 = p1+x2;
  if initial() then
    der(x1) = p1;
  else
    der(x1) = -x1*time + p1;
  end if;
  der(x2) = p2*y1 + y2;
  when { x2 > 10, initial() } then
    n1 = pre(n1) + pre(n2) + 1;
    y1 = y2;
    b = pre(b) and not x2 > 10;
  end when;
  when y1 > y2 then
    n2 = pre(n2) - 1;
  end when;
initial equation
  der(x1) - 10*p2 = 0;
  pre(n1) = 4;
end Pregunta3;
```

3.a (1 punto) Clasifique las variables del modelo en variables de tiempo continuo y variables de tiempo discreto. Clasifique las variables de tiempo continuo en variables de estado y variables algebraicas. Explique cómo ha realizado las clasificaciones.

3.b (2 puntos) Escriba todas las variables a calcular en el problema de inicialización y el conjunto completo de ecuaciones del problema de inicialización. Indique si el número de variables y ecuaciones coincide. En caso afirmativo, ordene y resuelva las ecuaciones a fin de calcular el valor de todas las variables del problema de inicialización.

3.c (1 punto) Escriba la condición de disparo de cada una de las cláusulas when, indicando en cada caso cómo el algoritmo de la simulación de los entornos de modelado de Modelica detecta esos eventos durante la simulación.

Solución a la Pregunta 3

Apartado 3.a

p_1 y p_2 son parámetros del modelo.

Las variables n_1, n_2, b son de tiempo discreto, ya que las dos primeras son de tipo entero y la tercera Booleana.

La variable y_1 es de tiempo discreto, ya que se calcula dentro de una cláusula when. Se trata de una variable de tipo real.

Las variables reales x_1, x_2, y_2 son variables de tiempo continuo, ya que se calculan de las ecuaciones de tiempo continuo del modelo. El problema de tiempo continuo está descrito mediante las tres ecuaciones siguientes:

$$y_2 = p_1 + x_2 \tag{1.6}$$

$$\frac{dx_1}{dt} = -x_1 \cdot t + p_1 \tag{1.7}$$

$$\frac{dx_2}{dt} = p_2 \cdot y_1 + y_2 \tag{1.8}$$

El problema de tiempo continuo es no singular. Tiene 3 ecuaciones y 3 incógnitas (y_2, $derx_1$, $derx_2$). Las dos variables que aparecen derivadas se seleccionan como variables de estado, siendo la siguiente una posible ordenación del modelo resuelto:

$$x_1,\ x_2 \quad \text{Variables de estado} \tag{1.9}$$

$$[y_2] = p_1 + x_2 \tag{1.10}$$

$$[derx_1] = -x_1 \cdot t + p_1 \tag{1.11}$$

$$[derx_2] = p_2 \cdot y_1 + y_2 \tag{1.12}$$

Apartado 3.b

Las variables del problema de inicialización pueden clasificarse de la forma siguiente:

Parámetros:	`p1`, `p2`
Variables de tiempo continuo:	`x1`, `x2`, `derx1`, `derx2`, `y2`
Variables de tiempo discreto:	`y1`, `n1`, `n2`, `b`,
	`pre(y1)`, `pre(n1)`, `pre(n2)`, `pre(b)`

La relación entre el código Modelica y las ecuaciones del problema de inicialización es la siguiente:

```
model Pregunta3
  parameter Real p1(start=0, fixed=false);  //
  parameter Real p2 = 11.2;                 // p2 = 11.2
  Real x1(start=-1, fixed=true);            // x1 = -1
  Real x2(start=3, fixed=false);            //
  Real y1(start=20, fixed=true);            // pre(y1) = 20
  Real y2(start=1.2, fixed=true);           // y2 = 1.2
  Integer n1;
  Integer n2(start=-10, fixed=true);        // pre(n2) = -10
  Boolean b(start=true, fixed=true);        // pre(b) = true
equation
  y2 = p1+x2;                               // y2 = p1+x2
  if initial() then
    der(x1) = p1;                           // der(x1) = p1
  else
    der(x1) = -x1*time + p1;
  end if;
  der(x2) = p2*y1 + y2;                     // der(x2) = p2*y1+y2
  when {x2>10, initial()} then
    n1 = pre(n1) + pre(n2) + 1;             // n1 = pre(n1)+pre(n2)+1
    y1 = y2;                                // y1 = y2
    b = pre(b) and not x2 > 10;             // b = pre(b) and not x2 > 10
  end when;
  when y1 > y2 then
    n2 = pre(n2) - 1;                       // n2 = pre(n2)
  end when;
initial equation
  der(x1) - 10*p2 = 0;                      // der(x1) - 10*p2 = 0
  pre(n1) = 4;                              // pre(n1) = 4
end Pregunta3;
```

Recopilando las ecuaciones anteriores, asignando la causalidad computacional y manipulando las ecuaciones, se obtiene el siguiente sistema de ecuaciones ordenado y resuelto del problema de inicialización, (existen otras posibles ordenaciones de las ecuaciones que son igualmente válidas):

```
p2 = 11.2
x1 = -1
pre(y1) = 20
y2 = 1.2
pre(n2) = -10
pre(b) = true
pre(n1) = 4
der(x1) - 10*p2 = 0         ->  der(x1) = 10*p2 = 112
der(x1) = p1                ->  p1 = der(x1) = 112
y2 = p1+x2                  ->  x2 = y2-p1 = 1.2-112 = -110.8
y1 = y2                     ->  y1 = 1.2
n1 = pre(n1)+pre(n2)+1      ->  n1 = 4-10+1 = -5
der(x2) = p2*y1+y2          ->  der(x2) = 11.2*1.2+1.2 = 14.64
b = pre(b) and not x2 > 10  ->  b = true and not -110.8>10 = true
n2 = pre(n2)                ->  n2 = -10
```

Apartado 3.c

Véanse las Secciones 4.3 y 4.6 del texto base.

Pregunta 4 (2 puntos)

Escriba un *script* en FlexPDE para obtener el mapa de contorno de distribución de temperatura y los vectores correspondientes al flujo de calor del sistema descrito a continuación. Se trata de un problema de dos dimensiones cuya geometría se muestra en la siguiente figura.



El sistema está formado por un círculo de un material de conductividad $\kappa_1 = 1$ W/(K·m) que tiene en su interior un triángulo de un material aislante de conductividad $\kappa_2 = 0.01$ W/(K·m). El círculo tiene de radio 0.4 m y su contorno exterior se mantiene a 300 K. El triángulo interior es equilátero, tiene de lado 0.2 m y su contorno exterior se mantiene a 230 K. La posición del triángulo en el interior del círculo se muestra en la figura.

Solución a la Pregunta 4

El *script* se muestra en el Código 1 y en las figuras siguientes se muestra el resultado de la simulación.

```
TITLE
   'ProblemaCalor'
SELECT
   errlim=1e-3     spectral_colors
VARIABLES
   temp
DEFINITIONS
   r2=0.4    k = 1
EQUATIONS
   div(-k* grad( temp))=0
BOUNDARIES
region 'medio1'
   start 'exterior' (0,-r2) value(temp)=300   { Exterior}
   arc( center=0,0)  angle=360
region 'medio2'
  k = 0.01
   start 'triangulo' (0,0)  value(temp)= 230  line to (-0.2,0) !Lado 1
   value(temp) = 230 LINE to (-0.1,0.17)    !Lado 2
   value(temp) = 230 LINE to close  !Lado 3
PLOTS
   contour(temp)       vector( -k*grad(temp))
END
```

Código 1: Modelo descrito en FlexPDE.


ProblemaCalor
10:52:34 11/23/25
FlexPDE Lite 6.51/W64
-k*grad(temp)
max 1.06
1.10
1.05
1.00
0.95
0.90
0.85
0.80
0.75
0.70
0.65
0.60
0.55
0.50
0.45
0.40
0.35
0.30
0.25
0.20
0.15
0.10
0.05
0.00
min 0.00
Scale = E3
Y
X
calor1: Grid#3 P2 Nodes=813 Cells=390 RMS Err= 0.0029


ProblemaCalor
10:52:34 11/23/25
FlexPDE Lite 6.51/W64
temp
max 300.
x : 300.
w : 297.
v : 294.
u : 291.
t : 288.
s : 285.
r : 282.
q : 279.
p : 276.
o : 273.
n : 270.
m : 267.
l : 264.
k : 261.
j : 258.
i : 255.
h : 252.
g : 249.
f : 246.
e : 243.
d : 240.
c : 237.
b : 234.
a : 231.
min 230.
Y
X
calor1: Grid#3 P2 Nodes=813 Cells=390 RMS Err= 0.0029
Integral= 140.4531